

Глава

3

Имена

В этой главе...

- ♦ Что такое имя
- ♦ Область действия имен
- ♦ Диспетчер имен
- ♦ Способы присвоения имен ячейкам и диапазонам
- ♦ Создание многолистных имен
- ♦ Работа с именами диапазонов и ячеек
- ♦ Поддержка имен ячеек и диапазонов
- ♦ Потенциальные проблемы с именами
- ♦ Ключ к пониманию термина “имя”
- ♦ Решение сложных задач с помощью формул

Большинство пользователей Excel знакомо с такими понятиями, как именованные ячейки и диапазоны. Их использование дает значительные преимущества. Прочитав эту главу, вы узнаете о различных типах имен, которые могут быть использованы в Excel. Возможно, вы и не подозреваете о том, как именованные ячейки смогут помочь вам в работе.

Что такое имя

Имя можно рассматривать как идентификатор какого-либо элемента рабочей книги. Имя присваивается таким элементам, как ячейка, диапазон, диаграмма, фигура и т.д. Имя, присвоенное, например, диапазону, может затем использоваться в формулах. Предположим, что диапазон B2 : B200 рабочего листа содержит информацию о дневном обороте компании. В ячейке C1 задана ставка комиссионного сбора с продажи. В следующей формуле сумма дневного оборота умножается на ставку комиссионного сбора:

=СУММ(B2 : B200) * C1

Такая формула будет работать, однако смысл ее не очевиден. Для того чтобы формула стала понятной, следует назначить описательные имена диапазону, содержащему ин-

формацию о дневном обороте, и ячейке, в которой указана ставка комиссионного сбора. Например, диапазону B2:B200 можно присвоить имя ДневныеПродажи, а ячейке C1 — имя СтавкаКомиссионных. Теперь формулу необходимо переписать, используя имена вместо адресов диапазона и ячейки.

=СУММ(ДневныеПродажи)*СтавкаКомиссионных

Как видите, формулу, в которой вместо ссылок на ячейки используются имена, понять значительно легче.

Применение именованных ячеек и диапазонов имеет ряд преимуществ.

- Имена облегчают понимание и применение формулы. Это особенно важно для тех, кто не участвовал в создании рабочего листа. Очевидно, что такую формулу, как, например, =Доходы-Налоги, понять значительно легче, чем =D20-D40.
- Описательное имя диапазона (например, Общий_доход) запоминается легче, чем адреса ячеек (например, A21). Кроме того, вероятность опечатки при написании имени меньше, чем при вводе адреса ячейки или диапазона.
- С помощью имен легко перемещаться по рабочему листу, используя поле Имя, расположенное слева строки формул. Щелкните на кнопке со стрелкой, и появится список всех назначенных имен. Перейти от одного элемента листа к другому можно, выбрав команду Главная⇒Редактирование⇒Найти и выделить⇒Перейти или нажав клавишу <F5>. В открывшемся диалоговом окне следует ввести имя диапазона или ячейки.
- Если выделить именованную ячейку или диапазон, ее (его) имя появится в поле Имя.
- Вам будет легче записывать формулы, используя имена ячеек и диапазонов. Эти имена можно вставить в формулу, выбрав команду Использовать в формуле во вкладке Формулы и выбрав в открывшемся списке нужное имя.
- Использование имен вместо адресов ячеек и диапазонов облегчит создание макросов и дальнейшую работу с ними.

Область действия имен

Перед тем как перейти к теме создания имен и работе с ними, важно осознать, что все имена имеют свою область действия. Под *областью действия* понимается множество мест, где может использоваться имя. В зависимости от области действия имена делятся на две категории.

- **Имена на уровне рабочей книги.** Могут использоваться в любом рабочем листе данной книги. Этот уровень присваивается именам по умолчанию.
- **Имена на уровне рабочего листа.** Могут использоваться только на рабочем листе, на котором они определены (если перед именем объекта не указано имя рабочего листа). В рабочей книге может быть много разных объектов с идентичными именами на уровне листа. Естественно, они должны находиться на разных листах.

Ссылки на имена

Ссылки на имена уровня книги позволяют применять их на любом листе книги. Ссылки на имена уровня листа должны предваряться именем соответствующего листа, если, конечно, используются на листе, отличном от того, на котором определены.

Для примера предположим, что рабочая книга содержит два листа: Лист1 и Лист2. В этой рабочей книге определены следующие имена: ОбщиеПродажи (имя уровня книги), ПродажиВЕвропе (имя уровня листа Лист1) и ПродажиВАмерике (имя уровня листа Лист2). На обоих листах книги можно обратиться к имени ОбщиеПродажи следующим образом:

=ОбщиеПродажи

На листе Лист1 к объекту ПродажиВЕвропе тоже можно обратиться только по имени, так как оно определено именно на этом листе.

=ПродажиВЕвропе

Однако, если на листе Лист1 нужно обратиться к имени ПродажиВАмерике, вам придется проделать немного больше работы. Лист1 не “видит” это имя, поскольку оно определено на другом листе. Лист1 может “видеть” только имена, определенные на нем. Для обращения к имени ПродажиВАмерике на листе Лист1 его придется предварить именем рабочего листа и восклицательным знаком.

=Лист2!ПродажиВАмерике



Совет

Если имя листа содержит пробел, заключите его в одинарные кавычки при обращении к имени, определенному на нем, например = 'Мой Лист' !МоеИмя.

В общем случае рекомендуется задавать именам как можно более узкую область действия. Если имя используется только на одном листе, ограничьте его область действия только этим листом. Если же имя используется на многих листах рабочей книги, более подходящей будет область действия уровня книги.



Примечание

В поле Имя отображаются только имена уровня текущего рабочего листа. Аналогично, только имена уровня текущего рабочего листа отображаются в списке команды Формулы⇒Определенные имена⇒Использовать в формуле.

Ссылка на имена другой рабочей книги

В главе 2 рассказывалось о том, как создавать ссылки на ячейки и диапазоны других рабочих книг. Те же правила применяются и к ссылкам на имена, заданные в другой рабочей книге.

Например, в следующей формуле используется имя ЕжемесячныеПродажи, определенное в рабочей книге под именем Бюджет.xlsx (предполагается, что эта рабочая книга открыта):

=СРЗНАЧ (Бюджет.xlsx!ЕжемесячныеПродажи)

Если имя ЕжемесячныеПродажи определено на уровне листа Лист1, последнюю формулу нужно записать так:

=СРЗНАЧ ([Бюджет.xlsx]Лист1!ЕжемесячныеПродажи)

Конфликты имен

Использование имен уровня листа может быть несколько запутанным, поскольку Excel разрешает на листах создавать те же имена, которые уже определены на уровне кни-

ги. В этом случае имена уровня листа имеют приоритет, но только в пределах листа, на котором определены.

К примеру, можно определить имя `Итог` уровня книги для некоторой ячейки листа `Лист1`. Одновременно можно определить имя уровня листа `Лист2!Итог`. Когда активен лист `Лист2`, идентификатор `Итог` будет ссылаться на имя уровня листа. Когда же активен какой-либо другой лист, `Итог` будет ссылаться на имя уровня книги. Запутанно, не правда ли? Для облегчения жизни рекомендуется избегать использования одних и тех же имен на уровнях книги и листа.

Один из способов устранения такой путаницы основан на соглашениях об именовании. Само имя должно содержать информацию о том, на каком уровне оно определено. К примеру, во всех именах уровня рабочей книги можно использовать префикс `рк`, а в именах уровня рабочего листа — `рл`. Используя это соглашение, вы никогда не спутаете имена `рлИтог` и `ркИтог`.

Диспетчер имен

Теперь, когда вы поняли концепцию области действия имен, можно приступить к их созданию и применению. Excel предлагает удобный способ управления именами — диспетчер имен (рис. 3.1).




Рис. 3.1. Диалоговое окно диспетчера имен

Для открытия окна диспетчера имен выберите команду `Формулы` ⇒ `Определенные имена` ⇒ `Диспетчер Имен` или нажмите клавиши `<Ctrl+F3>`. В этом диалоговом окне можно просматривать, создавать, редактировать и удалять имена. В главном окне диспетчера имен отображается текущее значение каждого имени, ячейка или диапазон, на которую оно ссылается, область действия имени, а также комментарии, если вы их создавали. Имена можно сортировать, а ширину столбцов изменять. Если используется много имен, можно применить к списку некоторый предопределенный фильтр, чтобы не загромождать список ненужными именами.

Следует обратить внимание, что размеры диалогового окна диспетчера имен можно изменять; перетаскивая за правый нижний угол, это диалоговое окно можно сделать шире или выше.

Создание имен

Диспетчер имен содержит кнопку **Создать**, при щелчке на которой открывается диалоговое окно, показанное на рис. 3.2.

В диалоговом окне создания имени можно задать имя, область действия и диапазон, на который ссылается имя. При желании можно сопроводить имя комментарием, который поможет впоследствии понять назначение имени. Поле **Диапазон** позволяет выбрать ячейки с помощью мыши или ввести диапазон вручную, аналогично тому, как это выполняется в строке формул.




Рис. 3.2. Диалоговое окно создания имени

Редактирование имен

При щелчке на кнопке **Изменить** откроется диалоговое окно **Изменение имени**, внешне идентичное окну **Создание имени**. В этом окне можно изменить любое свойство имени, за исключением области действия. При изменении имени диапазона все формулы рабочей книги, на него ссылающиеся, будут автоматически обновлены.



Совет

Для изменения области действия имя нужно вначале удалить, а затем воссоздать. Если при этом точно ввести то же имя, все формулы, его использующие, останутся работоспособными.

Диалоговое окно **Изменение имени** является не единственным местом редактирования имен. Если единственное свойство, которое нужно изменить, — диапазон, то можно сделать это непосредственно в главном окне диспетчера имен. В нижней части диалогового окна находится поле **Диапазон**. Выберите в списке нужное имя и измените ссылку в поле **Диапазон**.



Совет

При редактировании содержимого поля **Диапазон** вручную в строке состояния отображается слово **Укажите**. Если попытаетесь использовать при редактировании этого поля клавиши навигации, такие как стрелки, клавиши <Home> или <End>, то увидите, что перемещаетесь по рабочему листу, а не редактируете поле. Для переключения из режима указания в режим редактирования нужно нажать клавишу <F2>, в результате чего в строке состояния будет отображаться слово **Правка**.

Удаление имен

Щелчок на кнопке **Удалить** удаляет выделенное имя из рабочей книги. Естественно, Excel предупредит вас о том, что данную операцию нельзя будет отменить. К сожалению, программа недостаточно услужлива, чтобы заменить все ссылки на удаленное имя соответствующими диапазонами ячеек. В результате все формулы, использующие удаленное имя, будут отображать ошибку #ИМЯ?.

Способы присвоения имен ячейкам и диапазонам

В Excel существует несколько способов присвоения имен ячейкам и диапазонам, отличных от использования диспетчера имен. В данном разделе эти методы будут описаны. Кроме того, вы найдете и другую полезную информацию, имеющую отношение к именам.

Присвоение имен в окне Присвоение имени

Для того чтобы присвоить имя ячейке или диапазону, сначала выделите нужную ячейку или диапазон. Затем выберите команду **Формулы**⇒**Определенные имена**⇒**Присвоить имя**⇒**Присвоить имя**. Появится диалоговое окно, функционально и внешне идентичное окну **Создание имени**, открытому с помощью диспетчера имен (см. рис. 3.2).

Правила создания имен

Что касается назначаемых вами имен, Excel — достаточно неприхотливая программа. И все же существуют некоторые правила создания имен.

- **Имена не могут содержать пробелов.** Для разделения слов внутри имени можно использовать символ подчеркивания или точку (например, `Ежегодный_Итог` или `Ежегодный.Итог`).
- **В качестве имени можно использовать любую комбинацию букв и цифр.** Однако имя обязательно должно начинаться с буквы или символа подчеркивания. Имя не может начинаться с цифры (как, например, `3йКвартал`) или выглядеть как ссылка на ячейку (как, например, `К3`).
- **Для разделения слов внутри имени используются только символы подчеркивания и точки.** Этот момент не оговорен в документации к программе, однако Excel воспринимает и такие символы, как обратная косая черта (`\`) и вопросительный знак (`?`). Однако отмечу, что имя не может начинаться с этих символов.
- **Количество символов имени не должно превышать 255.** Этого вполне достаточно. Использовать имена такой длины не имеет смысла. Применение чрезмерно длинных имен лишает всякого смысла операцию присвоения имени диапазону.
- **В качестве имени можно использовать отдельные буквы (кроме 'R' и 'C').** Однако я бы не рекомендовал давать такие имена, поскольку имя должно нести информацию о содержимом ячейки или диапазона.
- **При вводе имен регистр значения не имеет.** `ЕжегодныйИтог` значит то же самое, что и `ежегодныйитог`. Excel сохраняет имя точно в таком же виде, в каком вы вводите его первоначально. Но, записывая имя в формуле, можно использовать любой регистр.

Excel использует несколько так называемых внутренних имен. Внутренние имена Excel можно присваивать ячейкам и диапазонам. Однако делать так не рекомендуется. Присваивая имя, избегайте следующих имен: `Область_печати`, `Заголовок_печати`, `Область_консолидации`, `База_данных`, `Критерий`, `Фильтр` и `Название_листа`.



Примечание

Количество имен ячеек и диапазонов не ограничено. Excel позволяет присваивать одному и тому же элементу сразу несколько имен, однако делать этого не следует. Если выделенная ячейка или диапазон имеет несколько имен, в поле **Имя** всегда будет появляться первое из них по алфавиту.

Имя можно присвоить и несмежному диапазону ячеек. Выделить все ячейки и диапазоны несмежного диапазона можно с помощью мыши, удерживая клавишу <Ctrl>.

Присвоение имени с помощью поля Имя

В поле **Имя** можно присвоить имя ячейке или диапазону значительно быстрее. Поле **Имя** — это раскрывающийся список, расположенный в левой части строки формул. Для того чтобы присвоить имя, выделите нужную ячейку или диапазон, затем щелкните в поле **Имя** и введите имя. Для подтверждения выбранного имени нажмите <Enter>. Если такое имя уже существует, то поле **Имя** нельзя использовать для другого диапазона или ячейки. Если попытаться сделать это, программа выделит исходный диапазон. Назначить существующее имя другому диапазону или ячейке можно только с помощью диалогового окна диспетчера имен.



Предупреждение

Создавая имя в поле **Имя**, не забудьте нажать <Enter>. Если ввести имя, а затем щелкнуть в области рабочего листа, новое имя не будет создано.

Поле **Имя** выполняет еще одну функцию. С его помощью можно быстро выделить именованную ячейку или диапазон, как показано на рис. 3.3. Для этого щелкните в поле **Имя** и выберите нужное имя в списке. Будет выделена соответствующая ячейка или диапазон. Как ни странно, для активизации поля **Имя** не предусмотрена комбинация клавиш. Иными словами, вы не получите доступ к этому полю с помощью клавиатуры; сделать это можно, только используя мышь. После щелчка в поле **Имя** для указания имени достаточно использовать клавиши со стрелками и <Enter>.

	B	C	D	E	F
n_a1					
n_a2	63,75p.				
Входящий Диапазон			2010		
год					
Доход					
Затраты					
квартал1	12				
Оплата	23				
7	65	23			
8	25	44			
9	24	34			

Рис. 3.3. Используя поле **Имя**, можно быстро активизировать именованную ячейку или диапазон



Совет

Имена, созданные в поле **Имя**, по умолчанию имеют область действия уровня рабочей книги. Для того чтобы создать имя уровня рабочего листа, предварите его именем рабочего листа и восклицательным знаком (например, Лист2!Итого). Так как поле **Имя** работает только с текущим выделенным диапазоном, ввод имени листа, отличного от текущего, приведет к ошибке.

Автоматическое присвоение имен

Содержимое ячеек можно использовать для автоматического создания имен смежных ячеек или диапазонов. На рис. 3.4 показан пример такого листа. Текст, находящийся

в столбце А, можно использовать для создания имен ячеек столбца В. Решить эту задачу в Excel достаточно просто.

	А	В	С	Д
3	Месяц	Продажи		
4	Январь	60 983,00р.		
5	Февраль	61 960,00р.		
6	Март	62 913,00р.		
7	Апрель	63 557,00р.		
8	Май	64 130,00р.		
9	Июнь	65 285,00р.		
10	Июль	66 449,00р.		
11	Август	67 302,00р.		
12	Сентябрь	68 065,00р.		
13	Октябрь	68 779,00р.		
14	Ноябрь	68 982,00р.		
15	Декабрь	70 789,00р.		
16				

Рис. 3.4. Excel позволяет создавать имена на основе текста, содержащегося в смежных ячейках

Для того чтобы присвоить имена, используя текст смежных ячеек, выделите данный текст и ячейки, которым должно быть присвоено имя (это могут быть как отдельные ячейки, так и диапазоны ячеек). Текст имен должен находиться в смежных ячейках относительно ячеек, которым вы хотите назначить имена. Можно создать несколько выделений. Выберите команду **Формулы**⇒**Определенные имена**⇒**Создать из выделенного** или нажмите комбинацию клавиш <Ctrl+Shift+F3>. Откроется диалоговое окно **Создание имен из выделенного диапазона** (рис. 3.5).




Рис. 3.5. Автоматическое создание имен

Флажки в диалоговом окне расставлены на основе проведенного программой анализа выделенного диапазона. Например, если Excel обнаруживает текст в первой строке выделенного диапазона, программа предлагает создать имена в строке выше. Если Excel приняла неправильное решение, измените предложенный вариант вручную. Щелкните на кнопке **ОК**, и Excel создаст имена. Обратите внимание на следующий факт: когда Excel создает имена, используя текст смежных ячеек, ячейки с текстом не включаются в именованный диапазон.

Если в результате использования текста смежной ячейки получится некорректное имя, Excel изменит его на правильное. Например, если ячейка содержит текст **Общий доход** (его использование в качестве имени некорректно, поскольку присутствует пробел), Excel преобразует пробел в символ подчеркивания и создаст имя **Общий_доход**.

Если же вместо текста смежная ячейка содержит значение или формулу, программа не сможет создать корректное имя. В таком случае Excel просто не создаст никакого имени.



Предупреждение

Внимательно проверяйте имена, автоматически создаваемые программой. Иногда диалоговое окно Создание имен из выделенного диапазона работает не совсем логично. На рис. 3.6 показана небольшая таблица с текстом и значениями. Выделите всю таблицу, выберите команду Создать из выделенного и примите вариант, предложенный программой (флажки в строке выше и в столбце слева). В результате имя **Товары** будет относиться не к диапазону A2 : A6, как вы ожидали, а к диапазону B2 : C6. Если верхняя левая ячейка выделенного диапазона содержит текст и установлены флажки в строке выше и в столбце слева, Excel использует это имя для всего набора данных, исключая верхнюю строку и левый столбец. Поэтому, прежде чем принять предложения программы, убедитесь, что она правильно поняла ваш замысел.




Рис. 3.6. Создание имен на основе данных такой таблицы может привести к неожиданным результатам

Имена столбцов и строк

Иногда имеет смысл назначить имя целому столбцу или строке. Нередко рабочий лист используется для хранения информации, которая вводится не каждый день. На рис. 3.7 приведен пример такого листа. Если создать имя не для всего столбца B, то придется изменять ссылку на диапазон, которому назначено это имя, каждый раз, когда вводятся новые данные. В таком случае целесообразно присвоить имя целому столбцу.

	A	B	C	D
1	Дата	Продажи		
2		янв.01	783	
3		янв.02	1 023	
4		янв.03	3 044	
5		янв.04	2 044	
6		янв.05	3 569	
7		янв.06		
8		янв.07		
9		янв.08		
10		янв.09		
11		янв.10		

Рис. 3.7. В этой таблице, содержащей данные о дневном объеме продаж, используется именованный диапазон, который состоит из целого столбца



Компакт-диск

Все приводимые в настоящей главе примеры содержатся в файле `Chapter3_RUS.xlsx` на прилагаемом компакт-диске.

Столбец В можно назвать, например, `ДневныеПродажи`. Если этот диапазон расположен на рабочем листе `Лист3`, ссылка будет выглядеть следующим образом:

```
=Лист3!$В:$В
```

Чтобы присвоить имя целому столбцу, выделите его, щелкнув на его букве. Введите нужное имя в поле **Имя** и нажмите клавишу `<Enter>`. Можно также применить для этого диалоговое окно **Создание имени**.

Назначив имя, можно использовать его в формуле. Следующая формула, например, возвращает сумму всех значений столбца В:

```
=СУММ(ДневныеПродажи)
```

Имена, созданные программой Excel

Excel создает и свои собственные имена. Например, если задать область печати для листа, Excel создаст имя `Область_печати`. Если назначить повторяющиеся строки или столбцы, которые должны быть напечатаны, программа создаст имя `Заголовки_печати` на уровне рабочего листа. Когда выполняется запрос, который возвращает данные в рабочий лист, Excel назначает этим данным свои имена. Ряд надстроек, входящих в пакет Excel, создает скрытые имена.

Вы вправе изменить ссылки на любые имена, автоматически создаваемые программой. Однако следует четко представлять себе, к чему эти изменения приведут.

Скрытые имена

Некоторые макросы и надстройки Excel создают скрытые имена. Эти имена присутствуют в рабочей книге, но не появляются в диалоговом окне диспетчера имен и в поле **Имя**. Надстройка "Поиск решения", например, создает целый ряд скрытых имен. Как правило, скрытые имена можно игнорировать, но они могут стать причиной возникновения некоторых проблем. Если скопировать лист из одной рабочей книги в другую, вместе с листом копируются и скрытые имена. При этом скрытые имена могут создавать ссылку, которую довольно трудно обнаружить.

К сожалению, в окне диспетчера имен скрытые имена не приводятся. Ниже приведена простая процедура на VBA, которая создает список всех скрытых имен активной рабочей книги. Эта процедура добавляет в книгу новый рабочий лист и помещает в него список скрытых имен. Не забудьте включить поддержку макросов. Кроме того, если список не появился, попробуйте удалить инструкцию `If`. Возможно, причина в том, что скрытых имен в данной книге нет.

```
Sub ListHiddenNames()
    Dim n As Name, r As Long
    Worksheets.Add
    r = 1
    For Each n In ActiveWorkbook.Names
        If Not n.Visible Then
            Cells(r, 1) = n.Name
            Cells(r, 2) = "" & n.RefersTo
            r = r + 1
        End If
    Next n
End Sub
```

Создание многолистовых имен

Имена ячейкам и диапазонам могут быть присвоены в нескольких рабочих листах с помощью “трехмерной” ссылки. Но для этого нельзя просто выделить диапазон и ввести имя в поле **Имя**. Чтобы назначить имя в нескольких листах, необходимо использовать диалоговое окно **Создание имени**. Формат такой ссылки будет выглядеть следующим образом:

ПервыйЛист: ПоследнийЛист! СсылкаНаДиапазон

На рис. 3.8 продемонстрирован пример, в котором имя **КубДанных**, назначенное диапазону **A1 : C3**, распространяется на листы **Лист1**, **Лист2** и **Лист3**.




Рис. 3.8. Присвоение имени в нескольких листах

Разумеется, можно просто ввести трехмерную ссылку на диапазон в поле **Диапазон**. Но если нужно создать имя, отметив диапазон мышью, могут возникнуть некоторые трудности. Даже если попытаться выделить диапазон, распространяющийся на несколько рабочих листов, Excel не сможет использовать адрес выделенного диапазона в диалоговом окне **Создание имени**.

Для того чтобы назначить имя **КубДанных** диапазону **A1 : C3**, который распространяется на листы **Лист1**, **Лист2** и **Лист3**, следуйте приведенным ниже инструкциям.

1. Активизируйте лист **Лист1**.
2. Во вкладке **Формулы** выберите команду **Диспетчер имен** или нажмите комбинацию клавиш **<Ctrl+F3>**. В открывшемся диспетчере имен щелкните на кнопке **Создать**. Откроется диалоговое окно **Создание имени**.
3. Введите **КубДанных** в текстовом поле **Имя**.
4. Активизируйте поле **Диапазон** и нажмите клавишу ****, чтобы удалить существующую ссылку на диапазон.
5. Выделите диапазон **A1 : C3** на листе **Лист1**. В поле **Диапазон** появится следующая запись:
=Лист1!\$A\$1:\$C\$3
6. Удерживая нажатой клавишу **<Shift>**, щелкните на вкладке **Лист3**. Вы заметите, что программа по непонятным причинам заменяет ссылку на диапазон ссылкой на одну ячейку. На данном этапе запись будет выглядеть так:
='Лист1:Лист3'!\$A\$1

7. Выделите диапазон A1 : C3 на листе Лист1 еще раз. На этот раз в поле **Диапазон** появится следующая запись:

= 'Лист1 : Лист3 ' ! \$A\$1 : \$C\$3

8. Теперь, когда поле **Диапазон** содержит нужную ссылку, щелкните на кнопке **ОК**. Диалоговое окно **Создание имени** будет закрыто.

Задав имя, можно использовать его в формулах. Например, следующая формула вернет сумму значений диапазона, который носит имя **КубДанных**:

=СУММ(КубДанных)



Примечание

Многолистовые имена не отображаются ни в поле **Имя**, ни в диалоговом окне **Переход** (это окно отображается при выборе команды **Главная** ⇒ **Редактирование** ⇒ **Найти и выделить** ⇒ **Перейти**). Иными словами, Excel позволяет назначать такие имена, но вы не сможете автоматически выделить ячейки, к которым они относятся. Тем не менее, многолистовые имена отображаются в списке автозавершения при вводе формул.

Если вставить новый лист в рабочую книгу, в которой используются имена, относящиеся к нескольким листам, имена будут распространяться на новый лист только в том случае, если этот лист находится между первым и последним листом, заданным в определении имени. В предыдущем примере рабочий лист, вставленный между Лист1 и Лист2, оказывается в диапазоне **КубДанных**. Лист, вставленный перед Лист1 или после Лист3, не включается в диапазон.

Если удалить первый или последний лист, включенный в определение имени, Excel автоматически изменит запись в поле **Диапазон**. В рассматриваемом нами примере удаление Лист1 приведет к следующим изменениям в поле **Диапазон** имени **КубДанных**:

= 'Лист2 : Лист3 ' ! \$A\$1 : \$C\$3

Областью действия имен, распространяющихся на множество листов, должна быть рабочая книга. Если они определены на уровне рабочего листа, то будут функционировать корректно, однако в диалоговом окне диспетчера имен будет отображаться ошибка.

Работа с именами диапазонов и ячеек

Создав имена диапазонов и ячеек, можно выполнять над ними целый ряд операций. В этом разделе вы узнаете о том, как выполнять основные операции над именами диапазонов и ячеек.

Создание списка имен

Если вы создали достаточно большое количество имен, может возникнуть необходимость наглядно отобразить, к какому диапазону относится каждое из них. В дальнейшем это поможет выявить ошибки или документировать свою работу.

Вероятно, вы захотите создать список всех имен (и соответствующих им адресов) рабочей книги. Для того чтобы создать такой список, сначала выделите ячейку в пустой области рабочего листа (если создать список имен, состоящий из двух столбцов, то он будет отображаться поверх находящейся в ячейках информации), а затем выберите команду **Формулы** ⇒ **Определенные имена** ⇒ **Использовать в формуле** ⇒ **Вставить имена** (или нажми-

те клавишу <F3>). Откроется диалоговое окно **Вставка имени** (рис. 3.9), в котором будут перечислены все заданные имена. Для того чтобы вставить список имен, щелкните на кнопке **Все имена**.



Предупреждение

В список имен не входят имена уровня рабочего листа (кроме имен активного листа), а также скрытые имена.

Список имен, вставленный в рабочий лист, состоит из двух столбцов. В первом столбце перечислены имена, а во втором — адреса соответствующих диапазонов. Адреса диапазонов, расположенные во втором столбце списка, представляют собой строки текста, которые напоминают формулы. Преобразовать такую строку в формулу можно, отредактировав ячейку (нажмите <F2>, а затем <Enter>). В результате строка становится формулой. Если имя относится к отдельной ячейке, формула отображает текущее значение этой ячейки. Если же имя относится к диапазону, формула возвращает ошибку #ЗНАЧ!.




Рис. 3.9. Диалоговое окно **Вставка имени**



Перекрестная ссылка

Ошибки в формулах, такие как #ЗНАЧ! и #ССЫЛКА!, рассматриваются в главе 21.

Использование имен в формулах

Задав имя ячейки или диапазона, можно использовать его в формулах. Следующая формула, например, вычисляет сумму значений диапазона **ПроданоЕдиниц**:
`=СУММ(ПроданоЕдиниц)`

Когда имя, заданное на уровне рабочего листа, используется в формуле того листа, в котором оно было задано, имя рабочего листа не следует указывать перед именем диапазона. Если такое имя используется в другом листе рабочей книги, необходимо задать полное имя (имя рабочего листа, восклицательный знак и имя диапазона). Например, если **ПроданоЕдиниц** — имя, заданное на уровне рабочего листа **Лист1**, то формула для вычисления суммы диапазона на любом другом рабочем листе будет выглядеть так:
`=СУММ(Лист1!ПроданоЕдиниц)`

Ввод знака равенства активизирует функцию автозавершения, как при вводе открывающей скобки в функции рабочего листа. При этом под активной ячейкой открывается список определенных имен и функций рабочего листа. Для использования функции автозаполнения вводите имя, пока оно не будет выделено в списке, а затем нажмите <Tab>, чтобы зафиксировать имя. Можно также использовать клавишу <↓>, чтобы выбрать имя в списке.

Если в формуле используется несуществующее имя, Excel отобразит ошибку #ИМЯ?, указывающую на то, что программа не может найти имя, которое вы пытаетесь использовать. Зачастую причиной такой ошибки является просто опечатка.

Использование операторов пересечения

В качестве оператора пересечения диапазонов в Excel служит единичный символ пробела. Например, следующая формула используется для вычисления суммы ячеек на пересечении двух диапазонов — B1 : C20 и A8 : D8:

=СУММ (B1 : C20 A8 : D8)

Данные диапазоны пересекаются в ячейках B8 и C8.

Оператор пересечения работает и с именованными диапазонами. На рис. 3.10 показана таблица, содержащая именованные диапазоны для столбцов и строк таблицы. Например, имя Январь относится к диапазону B2 : E2, а имя Север — к диапазону B2 : B13. Следующая формула возвращает содержимое ячейки на пересечении диапазонов Январь и Север:

=Январь Север

	A	B	C	D	E	F	G	H
1		Север	Юг	Запад	Восток			
2	Январь	235,00р.	463,00р.	118,00р.	325,00р.			
3	Февраль	465,00р.	350,00р.	406,50р.	262,25р.			
4	Март	256,00р.	360,50р.	355,25р.	380,88р.			
5	Апрель	452,00р.	354,00р.	357,25р.	356,25р.			235,00р.
6	Май	495,00р.	473,50р.	413,75р.	385,50р.			
7	Июнь	275,00р.	385,00р.	429,25р.	421,50р.			
8	Июль	632,00р.	453,50р.	419,25р.	424,25р.			
9	Август	457,00р.	544,50р.	499,00р.	459,13р.			
10	Сентябрь	421,00р.	439,00р.	491,75р.	495,38р.			
11	Октябрь	452,00р.	436,50р.	437,75р.	464,75р.			
12	Ноябрь	652,00р.	552,00р.	494,25р.	466,00р.			
13	Декабрь	573,00р.	612,50р.	582,25р.	538,25р.			
14								

Рис. 3.10. Эта таблица содержит именованные диапазоны, в которых заданы названия для столбцов и строк

Используя символ пробела для разделения двух ссылок на диапазоны, вы задаете так называемое *явное пересечение*. Это означает, что вы приказываете Excel явно определить пересечение диапазонов. Программа может находить и *неявные пересечения*. Неявное пересечение происходит в том случае, когда Excel выбирает значение из многоячеечного диапазона, основываясь на строке или столбце формулы, имеющей ссылку. Для того чтобы разобраться с пересечениями такого типа, рассмотрим конкретный пример. На рис. 3.11 показана таблица, содержащая именованный диапазон B3 : B8. В ячейке D5 отображается простая формула, приведенная ниже:

=МоиДанные

Обратите внимание на то, что в ячейке D5 отображено значение из диапазона МоиДанные, соответствующее строке, в которой расположена формула. Аналогично, если ввести ту же формулу в строки с 3 по 8, формула вернет соответствующее значение из диапазона МоиДанные. Excel выполняет неявное пересечение, используя диапазон МоиДанные и строку, содержащую формулу. Это действие подобно вычислению следующей формулы:

=МоиДанные 5 : 5

Если ввести формулу в строку, не принадлежащую диапазону МоиДанные, то формула вернет ошибку, поскольку неявное пересечение в таком случае произойти не может.

	A	B	C	D	E	F
1						
2						
3		6,00				
4		12,00				
5		18,00		18,00		
6		24,00				
7		36,00				
8						
9						

Рис. 3.11. Диапазон В3 : В8 этого рабочего листа имеет имя МоиДанные. Ячейка D5 демонстрирует неявное пересечение

Между прочим, неявные пересечения работают не только с именованными диапазонами. В рассматриваемом нами примере мы получим тот же результат, если в ячейке D5 будет находиться следующая формула (в которой не используется имя диапазона):

= $\$B\$2 : \$B\8

Если применить имя МоиДанные в качестве аргумента функции, неявное пересечение вычисляется, только когда аргумент состоит из одного значения. Например, если ввести приведенную ниже формулу в ячейку D3, произойдет неявное пересечение и формула вернет значение 64 (т.е. два в шестой степени).

=СТЕПЕНЬ (2 ; МоиДанные)

Но если в эту же ячейку ввести следующую формулу, неявного пересечения не будет, и формула вернет сумму всех значений диапазона МоиДанные:

=СУММ (МоиДанные)

Использование оператора диапазона

Работая с именованными диапазонами, можно использовать оператор диапазона, в качестве которого в Excel применяется символ двоеточия (:). Вернемся к примеру на рис. 3.10. Следующая формула возвращает сумму значений ячеек в прямоугольнике, состоящем из столбцов от Север до Запад и строк от Январь до Март:

=СУММ ((Север Январь) : (Запад Март))

Ссылка на отдельную ячейку именованного диапазона

Для того чтобы вернуть значение ячейки диапазона, можно использовать функцию Excel ИНДЕКС. Предположим, диапазон A1 : A50 имеет название ДиапазонДанных. Следующая формула возвращает второе значение (значение ячейки A2) диапазона ДиапазонДанных:

=ИНДЕКС (ДиапазонДанных ; 2)

Второй и третий аргументы функции ИНДЕКС — необязательные, однако один из них всегда должен быть задан. Второй аргумент (используемый в приведенной формуле) обозначает смещение строки в диапазоне ДиапазонДанных.

Если диапазон ДиапазонДанных состоит из нескольких ячеек, расположенных в одной строке, используйте следующую формулу, в которой пропущен второй аргумент

функции ИНДЕКС, но задан третий аргумент, определяющий смещение столбца в диапазоне ДиапазонДанных.

=ИНДЕКС (ДиапазонДанных ; 2)

Если же диапазон состоит из нескольких строк и столбцов, необходимо применить оба аргумента функции ИНДЕКС. Следующая формула, например, возвращает значение ячейки, расположенной в четвертой строке и пятом столбце диапазона ДиапазонДанных:

=ИНДЕКС (ДиапазонДанных ; 4 ; 5)

Применение имен в уже существующих формулах

После создания имени ячейки или диапазона Excel не будет автоматически заменять именами ссылки на эту ячейку или диапазон в формулах, созданных ранее. Тем не менее можно приказать программе использовать новое имя в тех или иных формулах.

Для этого выделите диапазон, содержащий формулы, которые должны быть преобразованы. Затем выберите команду **Формулы**⇒**Определенные имена**⇒**Присвоить имя**⇒**Применить имена**. Откроется диалоговое окно **Применение имен** (рис. 3.12). В нем выберите имена, которые нужно использовать в формулах. В формулах будут использованы только те имена, которые вы указали.



Совет

Для того чтобы применить новое имя ко всем формулам рабочего листа, выделите одну ячейку, перед тем как выполнить команду **Применить имена**.

Флажок **Игнорировать тип ссылки** отвечает за то, как Excel заменяет адрес диапазона именем. Обычно имя ячейки или диапазона задается как абсолютная ссылка. Если флажок **Игнорировать тип ссылки** установлен, Excel применит имя только в том случае, если ссылка в формуле полностью соответствует типу диапазона. В большинстве случаев, когда вы применяете имена, удобнее игнорировать тип ссылки.

Если флажок **Использовать имена строк и столбцов** установлен, то при применении имен в формулах Excel использует оператор пересечения. Не имея возможности использовать точные имена ячеек, Excel для ссылок на ячейки использует имена строк и столбцов. Для того чтобы объединить имена, Excel использует оператор пересечения. Щелчком на кнопке **Параметры >>** можно задать несколько дополнительных параметров, но только при условии, что флажок **Использовать имена строк и столбцов** установлен.




Рис. 3.12. Диалоговое окно **Применение имен**

Автоматическое применение имен при создании формул

Если вставить в формулу ссылку на ячейку или диапазон, указав этот элемент таблицы мышью, Excel автоматически заменит адрес ячейки или диапазона именем (в том случае, если данному элементу присвоено имя).

Иногда это свойство программы очень помогает в работе. Но если вы захотите использовать адрес ячейки, а не ее имя, это свойство будет только мешать. К сожалению, отключить данную функцию невозможно. Если вы решили использовать адрес ячейки или диапазона, введите его вручную (не указывайте элемент таблицы мышью).

Отмена применения имени

В Excel не предусмотрен специальный метод отмены применения имени. Иными словами, вы не сможете автоматически заменить имя ячейки или диапазона обычной ссылкой на этот элемент таблицы. Однако существует прием, с помощью которого можно заменить имя ссылкой. Для этого нужно установить режим вычислений по правилам Lotus 1-2-3. Откройте окно параметров программы и перейдите во вкладку **Дополнительно**. В разделе **Совместимость с Lotus** установите флажок **Клавиши перемещения, использовавшиеся в Lotus 1-2-3**, после чего щелкните на кнопке **ОК**.

Затем нажмите <F2> для того, чтобы отредактировать формулу, содержащую одно или несколько имен ячеек или диапазонов. Нажмите <Enter>, и редактирование ячеек будет завершено. Затем вернитесь в диалоговое окно параметров программы и снимите флажок **Клавиши перемещения, использовавшиеся в Lotus 1-2-3**. Отредактированная таким образом ячейка вместо имен будет содержать относительные ссылки на диапазоны.



Примечание

Описанный выше прием не документирован и может работать не во всех случаях, так что внимательно проверяйте результат.

Имена с ошибками

Если удалить строку или столбец, содержащий именованный диапазон или ячейку, само имя как таковое удалено не будет. Вместо этого в имени будет содержаться некорректная ссылка. К примеру, если ячейка A1 листа Лист1 имеет имя Дисконт, а вы удалили столбец A или строку 1, имя Дисконт будет содержать ссылку =Лист1!#ССЫЛ! (т.е. ошибочную ссылку). Если при этом имя Дисконт используется в какой-либо формуле, результатом формулы тоже будет #ССЫЛ!.

Чтобы не было такой ошибки, нужно в диспетчере имен удалить имя. Кроме того, можно переопределить имя так, чтобы оно указывало на корректную ячейку.



Совет

Диспетчер имен позволяет фильтровать отображаемые имена, используя предопределенные фильтры. Один из этих фильтров специально предназначен для отображения только тех имен, которые имеют ошибочные ссылки. Это позволяет быстро локализовать проблемные имена.

Просмотр именованных диапазонов

Если изменить масштаб изображения до 39% и меньше, вокруг именованных диапазонов появятся имена и границы, отображенные синим цветом (рис. 3.13). На печать эти имена и границы не выводятся — они предназначены только для того, чтобы вы могли увидеть именованные диапазоны на экране.

	А	В	С	Д	Е	Г	Г	И	И	К	Л	М
1	571	205	529	200	479	146						
2	576	507	129	204	297	516						
3	567	300	279	297	297	101						
4	567	207	240	297	297	225						
5	581	556	379	436	517	477						
6	581	247	167	561	177	61						
7	527	40	165	264	119	40						
8	566	164	240	297	297	297						
9	420	167	269	207	507	507						
10	158	462	560	407	257	270						
11	567	125	115	229	564	290						
12												
13												
14												
15												
16												
17												
18												
19												
20												
21												
22												

Рис. 3.13. Отображение имен диапазонов

Использование имен в диаграммах

При создании диаграммы каждая последовательность данных имеет соответствующую формулу РЯД. Формула РЯД содержит ссылки на диапазоны, используемые в диаграмме. Если в диаграмме используется именованный диапазон, можно отредактировать формулу РЯД, заменив ссылку именем диапазона.



Перекрестная ссылка

Дополнительная информация о диаграммах приведена в главе 17.

Поддержка имен ячеек и диапазонов

В процессе редактирования рабочего листа Excel автоматически поддерживает созданные вами имена ячеек и диапазонов. В следующих примерах мы будем рассматривать рабочий лист Лист1, который содержит имя МойДиапазон, ссылающееся на диапазон из девяти ячеек и заданное на уровне рабочей книги:

```
=Лист1!$C$3:$E$5
```

Вставка строки или столбца

При вставке строки выше именованного диапазона или столбца слева от именованного диапазона Excel автоматически изменяет ссылку на этот диапазон так, чтобы она соответствовала новому адресу диапазона. Например, если вставить новую строку 1, имя МойДиапазон будет ссылаться на =Лист1!\$C\$4:\$E\$6.

Если вставить новую строку или столбец в именованный диапазон, этот диапазон расширится и будет включать в себя новые строки или столбцы. Например, если вставить новый столбец слева от столбца Е, имя МойДиапазон будет относиться к =Лист1!\$C\$3:\$F\$5.

Удаление строки или столбца

При удалении строки, расположенной выше именованного диапазона, или столбца слева от именованного диапазона Excel автоматически изменяет ссылку на этот диапазон так, чтобы она соответствовала новому адресу диапазона. Например, если удалить строку 1, имя МойДиапазон будет относиться к =Лист1!\$C\$2:\$E\$4.

Если удалить строку или столбец внутри именованного диапазона, этот диапазон изменится соответствующим образом. Например, если удалить столбец D, имя `МойДиапазон` будет относиться к `=Лист1!C3:D5`.

Если удалить все строки и ячейки, составляющие именованный диапазон, именованный диапазон будет продолжать существовать, однако ссылка на него будет содержать ошибку. Например, удалите столбцы C, D, E, и имя `МойДиапазон` будет ссылаться на `=Лист1!#ССЫЛКА!`. Все формулы, в которых использовалось это имя, тоже будут возвращать ошибки.

Вырезание и вставка именованного диапазона

Когда вы вырезаете и вставляете именованный диапазон целиком, Excel изменяет ссылку соответствующим образом. Например, переместите диапазон `МойДиапазон` в другую часть листа, начинающегося с ячейки A1, и имя `МойДиапазон` будет сопоставлено ссылке `=Лист1!A1:C3`. Если вырезать и вставить только часть именованного диапазона, ссылка не изменится.

Потенциальные проблемы с именами

Имена часто используются для создания таблиц. Однако некоторые проблемы могут возникать с ними и при работе в книге. В этом разделе речь пойдет о тех моментах, о которых вы должны помнить, работая с именами в рабочих книгах.

Проблемы, возникающие при копировании рабочих листов

Как вы, вероятно, знаете, Excel позволяет копировать рабочие листы в пределах рабочей книги, а также в другую рабочую книгу. Прежде всего, мы рассмотрим пример копирования рабочего листа в пределах рабочей книги. Если копируемый лист содержит имена, заданные на уровне рабочего листа, те же имена будут присутствовать и в копии этого листа, причем в ссылках используется имя нового рабочего листа. Это именно то, чего вы ожидаете. Если же рабочая книга содержит имя, заданное на уровне рабочей книги, и данное имя относится к ячейке или диапазону копируемого рабочего листа, то оно также будет присутствовать и в копии. Однако это имя будет преобразовано в имя, заданное на уровне рабочего листа (чего вы, скорее всего, не ожидаете).

Рассмотрим рабочую книгу, содержащую один лист (`Лист1`). В ней присутствует имя `ИмяКниги`, назначенное ячейке A1 на уровне рабочей книги. Кроме того, существует еще одно имя — `Лист1!ЛокальноеИмя`, назначенное ячейке A2 на уровне рабочего листа. Если создать копию листа `Лист1` в пределах рабочей книги, новый лист будет называться `Лист1 (2)`. Скопируйте лист, и вы обнаружите, что теперь рабочая книга содержит четыре имени (рис. 3.14).

Такое увеличение количества имен при копировании листа не только сбивает с толку, но и может привести к ошибкам, найти причину возникновения которых бывает непросто. Если в рассматриваемом нами примере ввести следующую формулу в скопированный лист, формула вернет содержимое ячейки A1 скопированного листа:

```
=ИмяКниги
```




Рис. 3.14. Копирование рабочего листа приводит к созданию дубликатов имен

Иными словами, используется вновь созданное на уровне рабочего листа имя (а не исходное имя, заданное на уровне рабочей книги).

Если скопировать лист рабочей книги, которая содержит имя диапазона, относящегося к нескольким листам, вместе с этим рабочим листом будет скопировано и имя диапазона. В поле **Диапазон** появится сообщение об ошибке **#ССЫЛКА!**.

Если же лист копируется в другую рабочую книгу, все имена исходной рабочей книги, относящиеся к ячейкам копируемого листа, будут скопированы в новую книгу. Это касается как имен, заданных на уровне рабочей книги, так и имен, заданных на уровне рабочего листа.



Примечание

Копирование и вставка ячеек из одного рабочего листа в другой не приводит к копированию имен даже в том случае, когда копируемый диапазон содержит именованные ячейки.

Какой можно сделать вывод? Копируя листы рабочей книги, содержащей имена, будьте очень внимательны. Скопировав лист, проверьте все имена и удалите те, которые были скопированы случайно.

Проблемы, возникающие при удалении рабочих листов

При удалении листа, который содержит ячейки, используемые в заданном на уровне рабочей книги имени, имя не удаляется. Оно остается в рабочей книге, но ссылка на ячейку или диапазон в поле **Формула** будет содержать ошибку.

На рис. 3.15 показано диалоговое окно диспетчера имен, в котором присутствует имя с ошибкой. Исходная рабочая книга содержала лист **Лист1** и ячейку **A1** с именем **ИмяКниги**, заданным на уровне рабочей книги. После удаления листа **Лист1** имя **ИмяКниги** осталось в рабочей книге. Однако в поле **Диапазон** диалогового окна **Присвоение имени** появилось следующее выражение:

=#ССЫЛКА!\$A\$1

Обычно ошибочные имена в рабочей книге не вызывают каких-либо проблем, и все же рекомендуется удалять или корректировать все имена, содержащие ошибочные ссылки.




Рис. 3.15. Удаление листа, содержащего диапазон Мой-Диапазон, приводит к появлению ошибочной ссылки

Именованние объектов

Когда вы добавляете на рабочий лист объект, присваиваемое ему по умолчанию имя отражает тип объекта (например, Прямоугольник1).

Чтобы изменить имя объекта, выделите нужный объект, введите новое имя в поле Имя и нажмите <Enter>. Исключение составляют диаграммы. В этом случае необходимо выделить диаграмму, и в контекстной вкладке Работа с диаграммами⇒Макет выбрать команду Свойства.

Поле Имя в Excel имеет некоторые особенности: с его помощью можно переименовать объект, однако список объектов в этом окне не отображается. Более того, Excel позволяет задавать диапазонам имена, уже присвоенные объектам, при этом несколько объектов могут иметь одно и то же имя. В диалоговом окне Диспетчер имен имена объектов не отображаются.

Ключ к пониманию термина “имя”

Пользователи Excel часто обращаются к *именованным ячейкам* и *именованным диапазонам*. В этой главе мы постоянно оперируем этими понятиями, однако данные термины не совсем точны.

Что мы подразумеваем под термином *имя*? Присваивая имя, вы в действительности создаете именованную формулу. В отличие от обычной формулы, именованная формула хранится не в ячейке, а в памяти Excel.

Если вы будете исходить из такой формулировки, вам будет легче овладеть сложными приемами, о которых пойдет речь в следующих разделах.

При работе в диалоговом окне диспетчера имен поле **Диапазон** содержит формулу, а поле **Имя** — ее имя. Вы уже заметили, что содержимое поля **Диапазон** всегда начинается со знака равенства, что свойственно формулам.

Как видно на рис. 3.16, рабочая книга содержит имя *Ставка*, присвоенное ячейке B1 листа *Лист1*. В поле **Диапазон** присутствует следующая запись:
=Лист1!\$B\$1

При обращении к имени *Ставка* Excel в действительности вычисляет формулу, которая носит это имя, и возвращает результат. Например, можно ввести в ячейку следующую формулу:

=Ставка*1,05

Оценивая эту формулу, Excel, прежде всего, анализирует формулу, которая носит имя Ставка (оно существует только в памяти программы, а не в ячейке). Затем Excel умножает результат именованной формулы на 1,05 и возвращает результат. Разумеется, приведенная выше формула равноценна формуле, в которой вместо имени используется ссылка на ячейку:

```
=Лист1!$B$1*1,05
```

Возникает вопрос: можно ли создать такую именованную формулу, которая бы не содержала ссылок на ячейки? Ответ на него вы найдете в следующем разделе.

Присвоение имени константе

Рассмотрим таблицу, с помощью которой создается счет и вычисляется сумма налога для указанного объема продаж. Ставку налога с оборота вставьте в ячейку и ссылку на эту ячейку используйте в различных формулах. Для того чтобы было понятно, что за величина указана в ячейке, присвойте этой ячейке имя СтавкаНалогов.

Существует еще один способ. На рис. 3.17 представлен результат выполнения следующих инструкций.

1. Выберите команду **Формулы** ⇒ **Определенные имена** ⇒ **Присвоить имя** ⇒ **Присвоить имя**. Откроется диалоговое окно **Создание имени**.
2. Введите имя (в данном случае СтавкаНалогов) в поле **Имя**.
3. Перейдите в поле **Диапазон**, удалите содержащуюся в нем запись и вместо нее введите простую формулу, такую как =0,075.
4. Щелкните на кнопке **ОК**, для того чтобы закрыть диалоговое окно.




Рис. 3.16. Формально имя Ставка представляет собой именованную формулу, а не именованную ячейку




Рис. 3.17. Присвоение имени константе

В результате выполнения приведенных инструкций создается именованная формула, в которой не используются ссылки на ячейки. Чтобы проверить все вышеизложенное на практике, введем в любую ячейку такую формулу:

```
=СтавкаНалогов
```

Эта простая формула возвращает 0,075 — результат именованной формулы СтавкаНалогов. Поскольку именованная формула всегда возвращает один и тот же результат, ее можно считать именованной константой. Эта константа может быть использована и в более сложных формулах, например:

```
=A1*СтавкаНалогов
```

Если не изменять область действия, установленную по умолчанию на уровень книги, то можно использовать константу `СтавкаНалога` на всех листах рабочей книги.

Присвоение имени текстовым константам

В предыдущем примере константа представляла собой числовое значение. Однако константа может также состоять из текста. Например, можно определить константу для названия компании. Для создания следующей формулы, которая называется `MS`, используется диалоговое окно **Создание имени**:

```
= "Microsoft Corporation"
```

Теперь в любой ячейке можно использовать такую формулу:

```
= "Годовой отчет: "&MS
```

Данная формула вернет текст `Годовой отчет: Microsoft Corporation`.



Примечание

Имена, не относящиеся к диапазонам, не отображаются в поле **Имя** и в диалоговом окне **Переход** (которое можно вызвать нажатием клавиши `<F5>`). И это логично, поскольку константы отсутствуют в ячейках рабочего листа. Константы отображаются в диалоговом окне **Вставка имени**, и это тоже логично, поскольку имена констант используются в формулах.

Как вы, вероятно, догадываетесь, значение константы может быть изменено в любой момент в диалоговом окне **Изменение имени**. Все, что нужно сделать, — это изменить значение в поле **Диапазон**. После того как диалоговое окно будет закрыто, Excel пересчитает все формулы, в которых фигурирует данная константа, на основе нового значения.

Описанный выше прием очень эффективен во многих ситуациях, однако изменение значения занимает некоторое время. Если константу разместить в ячейке, изменить ее значение будет значительно легче. Если значение действительно является константой, изменять его вам не придется.

Использование функций рабочего листа в именованных формулах

На рис. 3.18 приведен еще один пример именованной формулы. В данном случае формула будет называться `ТекущийМесяц` и иметь следующий вид:

```
=МЕСЯЦ(СЕГОДНЯ())
```

В формуле, показанной на рис. 3.16, были использованы две функции рабочего листа. Функция `СЕГОДНЯ` возвращает текущую дату, а функция `МЕСЯЦ` — номер месяца аргумента даты. Таким образом, вы можете ввести в ячейку следующую формулу, которая будет возвращать номер текущего месяца. Например, если текущий месяц — апрель, формула будет возвращать 4.

```
=ТекущийМесяц
```

Более эффективная именованная формула будет возвращать название месяца в текстовой форме. Такая формула, называемая `ИмяМесяца`, будет выглядеть следующим образом:

```
=ТЕКСТ(СЕГОДНЯ(); "MMMM")
```



Перекрестная ссылка

Детально функция `ТЕКСТ()` рассматривается в главе 5.

Теперь эту формулу можно ввести в ячейку, и она будет возвращать название текущего месяца в текстовой форме. В апреле формула будет возвращать текст Апрель.

`=ИмяМесяца`

Использование ссылок на ячейки и диапазоны в именованных формулах

На рис. 3.19 приведен еще один пример создания именованной формулы. На этот раз в формуле используется ссылка на ячейку. Эта формула, которая называется `ПервыйСимвол`, возвращает первый символ, содержащийся в ячейке A1 листа `Лист1`. Именованная формула выглядит так:

`=ЛЕВСИМВ(Лист1!A1;1)`




Рис. 3.18. Создание именованной формулы, в которой используются функции рабочего листа




Рис. 3.19. Создание именованной формулы, в которой используется ссылка на ячейку

Создайте такую именованную формулу, и вы сможете ввести ее в ячейку. Формула всегда возвращает первый символ, содержащийся в ячейке A1 листа `Лист1`.

`=ПервыйСимвол`

В следующей именованной формуле, которую мы рассмотрим, используется ссылка на диапазон. На рис. 3.20 показано диалоговое окно `Создание имени`, в котором задана именованная формула под именем `Итог`.

`=СУММ(Лист1!A1:D4)`

Создав такую именованную формулу, можно ввести ее в ячейку любого рабочего листа. Формула возвращает сумму значений ячеек диапазона A1:D4 листа `Лист1`.

`=Итог`

Обратите внимание на то, что в двух предыдущих формулах использовались абсолютные ссылки. По умолчанию все ссылки на ячейки и диапазоны, используемые в именованных формулах, являются абсолютными и включают название рабочего листа. Прочитайте следующий раздел, и вы узнаете, какие интересные именованные формулы можно создавать, применяя относительные ссылки.

Использование именованных формул с относительными ссылками

Как уже отмечалось, если для создания именованной формулы со ссылками на ячейки или диапазоны применяется диалоговое окно **Создание имени**, ссылки в поле **Диапазон** всегда абсолютны и включают название рабочего листа. В этом разделе речь пойдет об использовании в именованных формулах относительных ссылок на ячейки и диапазоны.

Использование относительной ссылки на ячейку

Начнем с простого примера. Для того чтобы создать именованную формулу с относительной ссылкой на ячейку, выполните следующее.

1. Откройте пустой рабочий лист.
2. Выделите ячейку A1 (этот шаг очень важен).
3. Откройте диалоговое окно **Создание имени**.
4. В поле **Имя** открывшегося окна введите **ЯчейкаСправа**.
5. Удалите содержимое поля **Диапазон** и введите следующую формулу (не указывая ячейку рабочего листа мышью):
`=Лист1!B1`
6. Щелкните на кнопке **ОК**, чтобы закрыть диалоговое окно **Создание имени**.
7. Введите в ячейку B1 необходимую информацию.
8. В ячейку A1 введите следующую формулу:
`=ЯчейкаСправа`

Вы обнаружите, что формула, расположенная в ячейке A1, возвращает содержимое ячейки B1.

Теперь скопируйте формулу ячейки A1 в несколько ячеек, расположенных под ней, и введите произвольные значения в столбец B. Вы увидите, что формула, находящаяся в столбце A, возвращает содержимое ячейки, расположенной справа. Иными словами, именованная формула **ЯчейкаСправа** имеет относительный характер.

Имя **ЯчейкаСправа** может быть использовано в любой ячейке (а не только в ячейках столбца A). Например, если ввести `=ЯчейкаСправа` в ячейку D12, эта ячейка будет возвращать содержимое ячейки E12.

Для того чтобы убедиться в том, что в именованной формуле **ЯчейкаСправа** действительно используется относительная ссылка на ячейку, активизируйте любую ячейку, кроме A1, и откройте диалоговое окно **Присвоение имени** (рис. 3.21). В списке выберите **ЯчейкаСправа** и обратите внимание на содержимое поля **Диапазон**. Вы увидите, что изменение формулы зависит от активной в данный момент ячейки. Например, если активна ячейка E5, формула в диалоговом окне будет выглядеть так:

`=Лист1!F5`

Если имя **ЯчейкаСправа** используется в другом рабочем листе, формула все равно будет ссылаться на ячейку, расположенную справа. Но это будет ячейка с соответствующим адресом на листе **Лист1**. Дело в том, что именованная формула содержит ссылку на рабочий лист. Для того чтобы формула правильно работала на любом рабочем листе, измените ее, следуя инструкциям.




Рис. 3.20. Создание именованной формулы, в которой используется ссылка на диапазон




Рис. 3.21. Формула под именем ЯчейкаСправа изменяется в зависимости от того, какая ячейка активна в данный момент

1. Активизируйте ячейку A1 рабочего листа Лист1.
2. Откройте диалоговое окно Диспетчер имен.
3. В списке диалогового окна диспетчера имен выберите элемент ЯчейкаСправа.
4. Удалите содержимое поля Диапазон и введите следующую формулу:
= !B1
5. Щелкните на ОК, чтобы закрыть диспетчер имен.

Измените ссылку таким образом, и вы убедитесь, что теперь формула под именем ЯчейкаСправа корректно работает во всех листах рабочей книги.



Примечание

Именованная формула ЯчейкаСправа не будет работать, если она используется в формуле, расположенной в столбце XFD, поскольку в этом случае формула ссылается на несуществующую ячейку (справа от столбца XFD столбцов больше нет).

Использование относительной ссылки на диапазон

Следующий пример основан на предыдущем. Речь пойдет о создании именованной формулы, которая возвращает сумму десяти ячеек, расположенных справа от определенной ячейки. Для того чтобы создать такую именованную формулу, следуйте приведенным ниже инструкциям.

1. Активизируйте ячейку A1.
2. Откройте диалоговое окно Создание имени.
3. В поле Имя открывшегося окна введите Сумма10Ячеек.
4. Введите следующую формулу в поле Диапазон:
=СУММ(!B1:!K1)

Создав именованную формулу, можно ввести следующую формулу в любую ячейку любого листа, и эта формула будет возвращать сумму десяти ячеек, расположенных справа:
=Сумма10Ячеек

Например, если ввести такую формулу в ячейку D12, она будет возвращать сумму значений диапазона E12:N12, состоящего из десяти ячеек.

Обратите внимание на то, что, когда вы создавали именованную формулу, активной была ячейка A1, поэтому относительные ссылки в определении формулы заданы относительно ячейки A1. Заметьте также, что в формуле не использовано имя рабочего листа. Отсутствие имени рабочего листа (при условии, что в формулу включен восклицательный знак) говорит о том, что эту формулу можно использовать на любом листе.

Если активизировать ячейку D12, а затем открыть диалоговое окно **Создание имени**, то в поле **Диапазон** для формулы под именем `Сумма10Ячеек` будет приведена следующая запись:

```
=СУММ(!E12: !N12)
```



Примечание

Именованная формула `Сумма10Ячеек` не будет работать, если ввести ее в ячейку, которая располагается в столбце, следующем за столбцом XET. В таком случае формула становится ошибочной, поскольку в ней появляется ссылка на несуществующую ячейку, расположенную за пределами столбца XFD.

Использование смешанных ссылок на диапазоны

Как уже отмечалось в главе 2, ссылка может быть абсолютной, относительной или смешанной. Смешанная ссылка на ячейку может состоять из следующих элементов:

- абсолютная ссылка на столбец и относительная ссылка на строку (например, `$A1`);
- относительная ссылка на столбец и абсолютная ссылка на строку (например, `A$1`).

Смешанные ссылки могут использоваться в именованных формулах. Применим такую ссылку на практике. Активизируйте ячейку B1. В диалоговом окне **Создание имени** создайте формулу под именем `ПерваяВСтроке`. Определение формулы должно выглядеть так:

```
=!$A1
```

В этой формуле использована абсолютная ссылка на столбец и относительная ссылка на строку. Следовательно, формула всегда возвращает значение, заданное в столбце A. Строка зависит от того, в какой ячейке введена формула. Например, если ввести следующую формулу в ячейку F12, она вернет значение ячейки A12:

```
=ПерваяВСтроке
```



Примечание

Формулу `ПерваяВСтроке` нельзя использовать в столбце A, поскольку будет создана циклическая ссылка. Формула в данном случае будет ссылаться сама на себя.

Решение сложных задач с помощью формул

В этом разделе будет приведено несколько примеров решения усложненных задач с использованием имен. Все эти примеры предполагают, что вы уже знакомы с приемами, описанными в предыдущих разделах.

Функция ДВССЫЛ и именованный диапазон

Функция `ДВССЫЛ` позволяет определять адрес ячейки косвенно. Например, если ячейка A1 содержит текст "C45", следующая формула будет возвращать содержимое ячейки C45:

```
=ДВССЫЛ(A1)
```

Функция ДВССЫЛ может использоваться и в работе с именованными диапазонами. На рис. 3.22 представлена таблица с 12 именами диапазонов, соответствующими названиям месяцев года. Например, имя Январь относится к диапазону В2:Е2. Ячейка В16 содержит такую формулу:

=СУММ(ДВССЫЛ(А16))

Эта формула возвращает сумму именованного диапазона, имя которого задано в текстовой форме в ячейке А16.

	A	B	C	D	E	F	G
1		Север	Юг	Запад	Восток		
2	Январь	235,00р.	463,00р.	118,00р.	325,00р.		
3	Февраль	465,00р.	350,00р.	406,50р.	262,25р.		
4	Март	256,00р.	360,50р.	355,25р.	380,88р.		
5	Апрель	452,00р.	354,00р.	357,25р.	356,25р.		
6	Май	495,00р.	473,50р.	413,75р.	385,50р.		
7	Июнь	275,00р.	385,00р.	429,25р.	421,50р.		
8	Июль	632,00р.	453,50р.	419,25р.	424,25р.		
9	Август	457,00р.	544,50р.	499,00р.	459,13р.		
10	Сентябрь	421,00р.	439,00р.	491,75р.	495,38р.		
11	Октябрь	452,00р.	436,50р.	437,75р.	464,75р.		
12	Ноябрь	652,00р.	552,00р.	494,25р.	466,00р.		
13	Декабрь	573,00р.	612,50р.	582,25р.	538,25р.		
14							
15							
16	Март	1 352,63р.					

Рис. 3.22. Использование функции ДВССЫЛ при работе с именованным диапазоном



Совет

Команду Данные⇒Работа с данными⇒Проверка данных можно использовать для вставки раскрывающегося списка в ячейку А16. В поле Тип данных открывшегося окна выберите пункт Список, а в поле Источник введите диапазон А2:А13. Теперь пользователь сможет выбрать название месяца из списка, а итоговое значение для месяца будет отображаться в ячейке В16.

Используя функцию ДВССЫЛ, можно ссылаться и на имена уровня рабочего листа. Предположим, что вы работаете с рабочими листами Область1, Область2 и т.д. Каждый лист содержит имя ИтогиПродаж, заданное на уровне рабочего листа. Следующая формула извлекает значение из соответствующего рабочего листа, используя имя листа, заданное в ячейке А1:

=ДВССЫЛ(А1&"!ИтогиПродаж")

Использование функции ДВССЫЛ для создания именованного диапазона с фиксированным адресом

В Excel можно создать имя, которое всегда ссылается на одну и ту же ячейку или диапазон даже в том случае, если вы вставите новые строки или столбцы. Предположим, нужно создать диапазон под именем ЛевыйВерхний, который всегда ссылается на ячейку А1. Если создать имя обычным способом, вставка новой строки 1 приведет к тому, что диапазон ЛевыйВерхний будет ссылаться на ячейку А2. Вставка нового столбца приведет к тому, что диапазон ЛевыйВерхний будет ссылаться на ячейку В2. Для соз-

дания именованного диапазона, который имеет фиксированный адрес, необходимо задать именованную формулу, содержащую в поле **Диапазон** следующее выражение:

```
=ДВССЫЛ (" $A$1 ")
```

Если создать такую формулу, диапазон **ЛевыйВерхний** будет ссылаться на ячейку A1 даже после вставки новых строк и столбцов. Функция ДВССЫЛ предыдущей формулы позволяет задать адрес ячейки косвенно с помощью текстового аргумента. Поскольку аргумент заключен в кавычки, он никогда не изменяется.

Использование массивов в именованных формулах

Массив — это набор элементов. Массив может быть вертикальным (состоящим из одного столбца) или горизонтальным (состоящим из одной строки). Массивом может быть также совокупность нескольких столбцов и строк.



Перекрестная ссылка

В части IV подробно описаны массивы и формулы массивов. Однако эта тема имеет отношение и к именам.

Задать массив можно с помощью фигурных скобок. Каждый элемент массива отделяется запятой или точкой с запятой. Запятая используется для разделения элементов, расположенных вертикально, а точка с запятой — для разделения элементов, находящихся горизонтально.

В диалоговом окне **Создание имени** можно создать формулу **ИменаМесяцев**, которая имеет такое определение:

```
={"Янв"; "Фев"; "Мар"; "Апр"; "Май"; "Июн"; "Июл"; "Авг";  
"Сен"; "Окт"; "Ноя"; "Дек" }
```

Эта формула определяет массив, который состоит из 12 текстовых строк, расположенных горизонтально.



Примечание

Вводя такую формулу, не забудьте о фигурных скобках. В диалоговое окно **Создание имени** формула массива вводится не так, как в ячейку.

Определив формулу **ИменаМесяцев**, ее можно использовать в других формулах. Однако в конечной формуле необходимо обозначить, какой из элементов массива используется. Для этого рекомендуется применить функцию **ИНДЕКС**. К примеру, следующая формула возвращает текст "Авг":

```
=ИНДЕКС (ИменаМесяцев ; 8)
```

Можно отобразить и весь массив, состоящий из 12 элементов, но для этого потребуются 12 смежных ячеек. Для того чтобы ввести 12 элементов массива в диапазон A1 : L1, следуйте приведенным ниже инструкциям.

1. В диалоговом окне **Создание имени** создайте формулу с именем **ИменаМесяцев**.
2. Выделите диапазон A1 : L1.
3. В строку формул введите **=ИменаМесяцев**.
4. Нажмите комбинацию клавиш <Ctrl+Shift+Enter>.

При нажатии <Ctrl+Shift+Enter> Excel вставляет формулу массива в выделенный диапазон. В данном случае формула будет введена в 12 смежных ячеек (рис. 3.23). Программа заключает формулу массива в скобки, напоминая вам, что это специфическая формула. В каждой из 12 ячеек диапазона A1 : L1 формула будет иметь следующий вид: {=ИменаМесяцев}

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	Янв	Фев	Мар	Апр	Май	Июн	Июл	Авг	Сен	Окт	Ноя	Дек	
2													
3													

Рис. 3.23. Именованную формулу, которая содержит массив из 12 элементов, можно разместить в 12 смежных ячейках

Создание формулы с динамическим размером

Динамическая формула — это формула, относящаяся к диапазону, размер которого не фиксирован. Вероятно, осмыслить новое понятие непросто. Рассмотрим пример.

Обратите внимание на таблицу, показанную на рис. 3.24. Эта таблица содержит сведения об объеме продаж за период с января по май.

Предположим, нужно присвоить имя ОбъемПродаж диапазону данных, расположенных в столбце В. При этом имя не должно относиться к пустым ячейкам. Иными словами, ссылка на диапазон ОбъемПродаж будет меняться каждый месяц, по мере ввода новых данных. Разумеется, можно изменять ссылку вручную каждый месяц с помощью диалогового окна **Создание имени**, однако гораздо эффективнее использовать динамическую формулу, которая будет автоматически меняться с каждым последующим вводом данных.

Прежде чем задать динамическую формулу, создайте таблицу, показанную на рис. 3.24. Затем выполните следующие действия.

	A	B	C	D
1	Январь	1 356,00р.		
2	Февраль	1 456,00р.		
3	Март	1 556,00р.		
4	Апрель	2 656,00р.		
5	Май	1 756,00р.		
6	Июнь			
7	Июль			
8	Август			
9	Сентябрь			
10	Октябрь			
11	Ноябрь			
12	Декабрь			
13				

Рис. 3.24. Динамическая формула может использоваться для ввода данных в столбец В

1. Откройте диалоговое окно **Создание имени**.
2. В поле **Имя** введите **ОбъемПродаж**.
3. В поле **Диапазон** введите следующую формулу:
=СМЕЩ(Лист1!\$B\$1;0;0;СЧЁТЗ(Лист1!\$B:\$B);1)
4. Щелкните на кнопке **ОК**, чтобы закрыть диалоговое окно **Создание имени**.

В результате выполнения этих действий создается формула, в которой используются функции СМЕЩ и СЧЁТЗ.



Примечание

В формуле предполагается, что диапазон не содержит пустых ячеек. Например, если ячейка B2 пустая, функция СЧЁТЗ пропустит ее и в результате функция СЧЁТ вернет неправильный диапазон.

Для того чтобы проверить, как работает эта формула, введите следующую формулу в любую ячейку, которая не расположена в столбце В:

=СУММ(ОбъемПродаж)

Эта формула вернет сумму значений столбца В. Обратите внимание на то, что ОбъемПродаж не отображается ни в поле Имя, ни в диалоговом окне Переход. Для того чтобы выбрать диапазон, откройте диалоговое окно Переход и введите ОбъемПродаж вручную.

Напрашивается вопрос: зачем все эти трудности? Ведь и простая формула, приведенная ниже, сделает все, что необходимо:

=СУММ(В:В)

Эффективность работы именованных формул с динамическим размером становится очевидной при создании диаграмм. С помощью таких формул можно создать диаграмму с наборами данных, которые автоматически изменяются при вводе новых сведений.



Перекрестная ссылка

Использование таблиц для хранения данных часто устраняет необходимость создания динамических диапазонов. Таблицы рассматриваются в главе 9.